



# Embedded Pattern Matching

Trevor L. McDonell

t.l.mcdonell@uu.nl

Utrecht University

Netherlands

Joshua D. Meredith

josh.meredith@iohk.io

IIOG

Australia

Gabriele Keller

g.k.keller@uu.nl

Utrecht University

Netherlands

## Abstract

Haskell is a popular choice for hosting deeply embedded languages. A recurring challenge for these embeddings is how to seamlessly integrate user defined algebraic data types. In particular, one important, convenient, and expressive feature for creating and inspecting data—pattern matching—is not directly available on embedded terms. We present a novel technique, *embedded pattern matching*, which enables a natural and user friendly embedding of user defined algebraic data types into the embedded language, and allows programmers to pattern match on terms in the embedded language in much the same way they would in the host language.

**CCS Concepts:** • **Software and its engineering** → **Functional languages; Domain specific languages; Data types and structures.**

**Keywords:** Haskell, pattern matching, algebraic data types, embedded languages

### ACM Reference Format:

Trevor L. McDonell, Joshua D. Meredith, and Gabriele Keller. 2022. Embedded Pattern Matching. In *Proceedings of the 15th ACM SIGPLAN International Haskell Symposium (Haskell '22)*, September 15–16, 2022, Ljubljana, Slovenia. ACM, New York, NY, USA, 14 pages. <https://doi.org/10.1145/3546189.3549917>

## 1 Introduction

Algebraic data types have proven to be a powerful, convenient, and expressive way of creating and organising data, and have become a defining characteristic of functional programming languages in particular. Algebraic data types and pattern matching are highly suited to defining abstract syntax trees and the operations on them. This enables a popular and time-saving way to define and implement a [domain-specific] programming language: by *embedding* it in another one. Embedding means to represent the terms and values of the *embedded language* as terms and values in the *host*

*language*, so that the former can be interpreted in the latter [30]. While a *shallow* embedding executes the operations of the embedded language directly, a *deep* embedding instead builds an abstract syntax tree (AST) that represents the embedded program. For example, the abstract syntax tree for the untyped lambda calculus can be defined as:

```
data Name = Name String
data Lambda = Var Name
            | App Lambda Lambda
            | Lam Name Lambda
```

Writing an evaluation function for terms in this language is a straightforward exercise, and more complex transformations such as optimisation and code generation are also possible.

The most light-weight (and probably most common) approach to obtaining such an AST is what we call a *combinator-based deep embedding*, where the language terms more or less directly call the constructors of the AST [4, 7, 11, 14, 31]. This allows the language implementor to use most of the host language compiler’s infrastructure. Depending on the expressivity of the host language type system, many of the sanity checks for the embedded language can also be delegated to the host language type checker. Using overloading, this type of deep embedding can be made almost completely transparent to the user, and gives the illusion of working in the host language directly.

Other popular approaches, which construct the AST by using template- or meta-programming [24, 36], or compiler plug-ins [6, 40] require more implementation effort, but in exchange offer more flexibility compared to the combinator-based approach as they reuse less functionality of the host language (for example, by implementing their own parser and type checker) and may manipulate the AST of the host language itself. One drawback of combinator-based deep embeddings, on the other hand, is that user-defined algebraic data types and pattern matching—the very features which make implementing the embedding itself so convenient—are themselves difficult to integrate into the embedding. Consequently, these important language features are not well supported, if at all.

In this paper we present a novel technique to support user-defined algebraic data types in combinator-based deep embeddings. While some embedded languages support product types to some extent, to the best of our knowledge there are no combinator-based deeply embedded languages which support sum or recursive data types, nor pattern matching on embedded values of algebraic data type. Our solution to

---

Permission to make digital or hard copies of part or all of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. Copyrights for third-party components of this work must be honored. For all other uses, contact the owner/author(s).

Haskell '22, September 15–16, 2022, Ljubljana, Slovenia

© 2022 Copyright held by the owner/author(s).

ACM ISBN 978-1-4503-9438-3/22/09.

<https://doi.org/10.1145/3546189.3549917>

the problem is based on two ideas. First, we automatically map algebraic data types to a generic internal representation which is amenable to inspection. Second, we develop a technique so that the pattern matching mechanism of the *host* language can be used to construct the abstract syntax for case distinctions in the *embedded* language. These techniques are sufficiently flexible to handle arbitrarily nested algebraic data types, including user-defined ones. Both the mapping to the internal representation and the pattern matching in the embedded language happen almost transparently to the user, so that the code they have to write to create, inspect, and manipulate values of algebraic data type in the embedded language closely resembles that of the host language. For example, we can write an *embedded* program that converts *embedded* lambda terms into SKI combinator form as:

```
toSKI :: Exp Lambda → Exp SKI
toSKI = match \case
  Var_ v   → ...
  App_ f x → ...
  Lam_ v x → ...
```

Our technique is not merely one of convenience: it also exposes optimisation opportunities which are obfuscated if the user has to resort to workarounds in the absence of these features. In this paper we present:


- a mapping for arbitrary user-defined algebraic data types to a generic internal representation in the abstract syntax tree of an embedded program (§4);
- embedded pattern matching, a user-friendly technique to lift pattern matching from the host language into the embedded language (§5); and
- a case study of these techniques implemented for an existing real-world embedded language, and discuss the impact of these changes both in terms of the user experience and efficiency of the generated code (§6).

We demonstrate our technique using a core expression language embedded in Haskell, but our technique is applicable to other host languages. Section 2 gives an overview of deep embeddings, and the problem algebraic data types pose in this setting. Section 3 briefly covers the syntax and semantics of pattern matching in Haskell. Section 7 discusses related work.

The source code for the implementation described in this paper is available at <https://github.com/tmcdonell/embedded-pattern-matching> [22].

## 2 Deeply Embedded Languages

The power of deeply embedded languages is that the implementer has complete control over the evaluation of the program and can support multiple interpretations of it. The disadvantage is that realising the embedding in a user-friendly way is much more challenging than in the shallow case, because we are essentially hiding from the user that they are



**Figure 1.** Execution stages of a deeply embedded program

working on syntax trees representing the computation of values, rather than computing on values directly.

Ideally, the user of an embedded language would be able to use all of the features of the host language in the embedding, and there is a long line of research addressing this problem [9, 11, 13, 21, 34]. Moreover, we would like the concrete syntax of the embedded language to be as close as possible to the equivalent expression in the host language.

Operator and syntax overloading help to make an embedding feel natural by providing implementations for primitive operations such as (+) or if-then-else on embedded terms. Higher-order embeddings, which make host language variables and abstraction available to the user, are also add convenience. The higher-order embedding can then be converted to a first-order embedding, again transparently to the user [9, 13, 21]. Many deeply embedded languages do not support algebraic data types at all, or only a fixed set of built-in data types. To understand why the challenges, in particular posed by the integration of algebraic data types, let us have a look some concrete examples. We start with the following Haskell function:

```
safeDiv :: Fractional a => a → a → Maybe a
safeDiv n d =
  if d == 0           -- operator; conditional
  then Nothing       -- constructor
  else let r = n / d -- operator; let-binding
        in Just r    -- constructor
```

Even though this function is quite simple, and only returns an algebraic data type as result, it already demonstrates much of the complexity of the task. An implementation of the function in an embedded language might have the type signature:

```
safeDiv' :: Fractional (Exp a)
          => Exp a → Exp a → Exp (Maybe a)
```

This embedded function does not take regular numeric values as input. Instead, it takes two *abstract syntax trees* of type `Exp a`, which represent a computation that—once evaluated—will result in a numeric value of type `a`. By reifying the program as an abstract syntax tree we have much more freedom in how we evaluate it. The evaluation function could simply be an interpreter in the host language, but it could also generate and compile code for the expression.

Fig. 1 shows an overview of the stages of evaluation for a language deeply embedded in Haskell. The key idea is that, when we write a program in a deeply embedded language, what we are really doing is writing a program in the *host*

language which, at host program *runtime*, will generate the abstract syntax for the embedded program ( $\text{Exp } a$ ), compile (or interpret) that program, execute it (possibly on a different computing device), and finally return the result of running the embedded program back into the host program. This cycle may be repeated many times during the execution of the host program.

This *staged compilation* means that values of type  $a$  are fundamentally different from values of type  $\text{Exp } a$ . While we can often *lift* values from the host to the embedded language by generating appropriate abstract syntax to represent that value, the reverse is not possible. The only way to return expression values back to the host is by evaluating the *entire* embedded program.

We use the colours of the different stages of program execution presented in Fig. 1 to provide a visual hint of where (or rather when) a value is available: either during execution of the **host** or **embedded** program. Effectively crossing between the different phases of embedded program execution is the key challenge solved by this paper.

Returning to our example, the challenges to cleanly implementing `safeDiv'` in a combinator-based deeply embedded language are, in order of increasing complexity:

**Type class overloading:** Operations in a deeply embedded language do not directly compute values. Instead, they combine the abstract syntax tree(s) of their argument(s) to form a new AST that represents the result of computing that operation. In Haskell, the division operator (`/`), for example, is provided by the `Fractional` type class. If the embedded language author provides an instance of this type class:

```
instance Fractional (Exp a) where
  (/) = ...      -- construct appropriate abstract syntax
```

then the (`/`) operation can be used transparently on embedded terms, and so programming in the embedded language feels a lot like programming in the host language.

**Conditionals:** Host language conditionals cannot be used directly on embedded Boolean expressions, since the condition needs to be of type `Bool`, not `Exp Bool`: the latter represents a calculation whose Boolean value cannot be determined at the time of generating the embedded program, only after that program is evaluated. Embedded languages often work around this limitation by instead providing a *function* that can be used to express embedded conditionals: `cond :: Exp Bool → Exp a → Exp a → Exp a`. If the host language supports overloading or rebinding of the built-in syntax to this expression language operation, then the traditional `if-then-else` syntax can be used in the embedded program as well.

**Variable bindings:** It is important to ensure that sharing of `let`-bound embedded expressions is preserved, and their value is not recomputed for every use of the bound variable.

Previous work has shown how to efficiently and conveniently use the host language mechanisms for variable binding and abstraction by using a higher-order embedding [27], which is then internally converted to a first-order embedding [1] while taking care to preserve [21] or recover [3, 13] any sharing. This conversion has to happen as the very first step when processing the AST to preserve sharing.

The techniques we describe in this paper are all concerned with the initial construction of the higher-order AST and do not interfere with sharing recovery.

**Algebraic data types:** Arguably, a functional language should support algebraic data types (ADTs), both user defined as well as built-in types such as `Maybe`. This means that we must address: (1) lifting algebraic data type values into the embedded language; and (2) constructing, inspecting, and decomposing embedded expressions on algebraic data types.

Lifting values of algebraic data types means the embedded language implementer must devise a mapping of the user-extensible set of *surface* types into a fixed set of *representation* types internally understood by the embedded language. The challenge with such a mapping is to do it in a way such that implementation details of the embedding do not leak to the user, for example via compiler error messages. Failing to do so can seriously affect the usability of the embedding.

The difficulty is in inspecting and decomposing abstract data type values in the embedded language, particularly for sum types. Consider the following function:

```
fromMaybe :: a → Maybe a → a
fromMaybe d Nothing = d
fromMaybe _ (Just x) = x
```

If we want to write the equivalent of this Haskell function in the embedded language, we unfortunately cannot pattern match directly on the second argument, for the same reason we could not use the built-in conditional operator: we will only know which constructor was passed as argument to the function once the expression has been evaluated; it is not available when generating the abstract syntax of the embedded language program. Conditional expressions can be seen as pattern matching specialised to Boolean values, where we were able to work around this limitation by using a function `cond` instead. Pattern matching, however, is more general and can therefore not be replaced by a single operator: for every pre-defined and user-defined algebraic data type, we need the functionality to create a term in the embedded language which expresses a case distinction to be made at embedded program execution time.

In this paper we show how we can lift the pattern matching functionality of the host language into the embedded language, and how to integrate this feature in a user-friendly way that is as close as possible to the syntax of pattern matching in the host language.

<code>var, x</code>	<code>::= ...</code>	Variable name
<code>con, K</code>	<code>::= ...</code>	Data constructor name
<code>expr</code>	<code>::= ...</code>	Expression
<code>pat</code>	<code>::= var</code>	Variable pattern
	<code>'_'</code>	Wildcard pattern
	<code>con pat<sub>1</sub> ... pat<sub>n</sub></code>	Constructor pattern
	<code>expr '→' pat</code>	View pattern

**Figure 2.** The syntax of patterns in Haskell

As we will see, the embedded version of the `fromMaybe` function can be implemented almost as the non-embedded version, only the names of the constructors in the pattern need to change:

```
fromMaybe' :: Exp a → Exp (Maybe a) → Exp a
fromMaybe' d Nothing_ = d
fromMaybe' _ (Just_ x) = x
```

Similarly, the full embedded implementation of the `safeDiv` function is also almost the same, apart from the name change:

```
safeDiv' :: Fractional (Exp a)
         ⇒ Exp a → Exp a → Exp (Maybe a)
safeDiv' n d =
  if d == 0           -- operator; conditional
  then Nothing_      -- constructor
  else let r = n / d  -- operator; let-binding
        in Just_ r    -- constructor
```

Before we discuss what happens behind the scenes to realise this embedding, we briefly revisit pattern matching in Haskell, as this will form the basis for our contribution.

### 3 Pattern Matching in Haskell

When describing our approach, we frequently refer to the syntax and semantics of pattern matching in Haskell [19] and we use its pattern synonyms extension [28] to make the integration between the host and embedded languages seamless. Therefore, we start with a brief summary of both.

Fig. 2 gives the core syntax of patterns in Haskell. Variable and constructor patterns are conventional, while view patterns are an extension to the basic syntax to allow computations to be performed during pattern matching. For example:

```
uncons  :: [a] → Maybe (a, [a])
safeHead :: [a] → Maybe a
safeHead (uncons → Just (x, _)) = Just x
safeHead _                       = Nothing
```

By using `uncons` inside a view pattern, matching the result of the view function against the `Just` constructor determines whether the first equation for `safeHead` succeeds or not.

#### 3.1 Pattern Synonyms

Pattern synonyms [28] are an extension to Haskell pattern matching that allows programmers to define new patterns. In an *implicitly bidirectional* pattern synonym the right-hand

<code>psyn</code>	<code>::= 'pattern' lhs '=' pat</code>	Implicitly bidirectional
	<code>'pattern' lhs '←' pat</code>	Explicitly bidirectional
	<code>'where' lhs '=' expr</code>	
<code>lhs</code>	<code>::= con var<sub>1</sub> ... var<sub>n</sub></code>	Prefix notation
<code>sig</code>	<code>::= 'pattern' con '::' pat_ty</code>	Pattern signature
<code>pat_ty</code>	<code>::= ...</code>	Pattern type

**Figure 3.** The syntax of pattern synonym declarations

side is used both as a pattern (when matching) and as an expression (when constructing). This works nicely because many Haskell patterns look like expressions, but, as we saw with view patterns, sometimes we want to do some *computation* at the same time. For example, suppose we want to convert between rectangular and polar coordinates:

```
data Cartesian = Cartesian Float Float
toPolar  :: Cartesian → (Float, Float)
fromPolar :: Float → Float → Cartesian
```

Rather than define a new datatype for polar coordinates and juggling both representations, we can convert between Cartesian and polar coordinates on demand by writing the following *explicitly bidirectional* pattern synonym:

```
pattern Polar :: Float → Float → Cartesian
pattern Polar r a ← (toPolar → (r, a))
  where Polar = fromPolar
```

There are two pieces to this declaration. First, the back-arrow “←” indicates that `Polar` can be used in patterns, wherein it consists of the view function `toPolar → (r, a)` to first compute the variables `r` and `a`, which are then bound on the left. Second, the `where` clause specifies how `Polar` should behave when used as an expression, in this case by converting the given polar coordinates into Cartesian space.

The definition in the `where` clause is the *builder*, while the pattern after the “←” is the *matcher*. Critically, the only required relationship between the builder and the matcher is that their types are compatible. One might expect that the builder and matcher have to be inverses of each other, as in `Polar`, but this is not the case. When we define our embedded pattern matching in Section 5 we will make use of this asymmetry, as both constructing *and* destructing embedded terms *adds* abstract syntax to an expression. A subset of the syntax of pattern synonyms is shown in Fig. 3.

#### 3.2 Semantics of Pattern Matching

In this section we summarise the semantics of pattern matching in Haskell [19, 28]. When matching against a value, there are three possible outcomes: the match *succeeds* (binding some variables); the match *fails*; or the match *diverges* (does not terminate or terminates with an error). Failure is not necessarily an error: if the match fails in a function defined by multiple equations, or a case expression with multiple alternatives, the next equation (or alternative) is tried.

$E$	$::=$	$C$	Constant
		$x$	Variable
		<b>let</b> $var = E$ <b>in</b> $E$	Let binding
		$\lambda x. E$	Abstraction
		$E E$	Application
		$T$	Tuples
		$\pi E$	Projection
		$trace E$	Pattern match
		$E [E_1 \dots E_n]$	Case expression
$T$	$::=$	$()$	Unit
		$E$	Expression
		$T T$	Pair
$\pi$	$::=$	$\dots$	Projections
$C$	$::=$	$\dots$	Constants
$var, x$	$::=$	$\dots$	Variable name
$trace$	$::=$	$\dots$	Match trace (§5.3)

**Figure 4.** The grammar of our embedded language

Matching a pattern  $p$  against a value  $v$  is given by cases which depend on the form of the pattern. If it is

1. a variable  $x$  the match succeeds, binding  $x$  to the value  $v$ .
2. a wildcard  $'_'$  the match succeeds, binding nothing.
3. a view pattern  $f \rightarrow p$  then evaluate  $f v$ . If evaluation diverges, the match diverges, otherwise match the resulting value against  $p$ .
4. of the form  $(P p_1 \dots p_n)$ , where  $P$  is a data constructor, or a pattern synonym defined by  $P x_1 \dots x_n = p$  or  $P x_1 \dots x_n \leftarrow p$ , then
  - a. if the value  $v$  diverges, the match diverges.
  - b. if the value is of the form  $(P' v_1 \dots v_m)$  where  $P \neq P'$  the match fails.
  - c. if the value is of the form  $(P v_1 \dots v_n)$  then we match the sub-patterns from left-to-right ( $p_1$  against  $v_1$  and so on). If any of the patterns fail (or diverge) then the whole pattern match fails (or diverges). The match succeeds if all sub-patterns succeed.

The main point to note is that for a pattern match to succeed, matching on the constructor or pattern synonym—as well as all of its sub-patterns—must succeed. The need to handle nested patterns complicates our task considerably. We will return to this point in Section 5.3.

### 3.3 Cosmetic Shortcomings of Pattern Synonyms

Using pattern synonyms for working with embedded data types provides a significant usability improvement for the embedded language user (§6). However, there are two cosmetic shortcomings of pattern synonyms in Haskell:

- Pattern synonyms exist in the same name space as data constructors, so we cannot use the same name for a constructor and its corresponding embedded pattern synonym. Our convention is to form the pattern synonym name by adding a trailing underscore to the constructor name.

```

data Exp a where
  Const  :: EltR a → Exp a
  Tuple  :: Tuple (TupleR t) → Exp t
  Prj    :: TupleIdx (TupleR t) e → Exp t → Exp e
  Roll   :: Exp a → Exp (Rec a)
  Unroll :: Exp (Rec a) → Exp a
  Match  :: TraceR (EltR a) → Exp a → Exp a
  Case   ::
    Exp a → [(TraceR (EltR a), Exp b)] → Exp b

data Tuple t where
  Unit   ::          Tuple ()
  Exp    :: Exp a      → Tuple a
  Pair   :: Tuple a → Tuple b → Tuple (a, b)

```

```

data TupleIdx s t where
  PrjZ  ::          TupleIdx t      t
  PrjL  :: TupleIdx l t → TupleIdx (l, r) t
  PrjR  :: TupleIdx r t → TupleIdx (l, r) t

```

```

newtype Rec a = Rec a

```

**Figure 5.** The core AST of our generic expression language

- It is not possible to overload built-in syntax, in particular for tuples. Instead we use the pattern synonym name  $T2$  for pairs  $(,)$ ,  $T3$  for triples  $(,,)$ , and so on.

## 4 Generic Expression Language

We explain our technique for embedding algebraic data types using the language shown in Fig. 4. It offers only a few operations: lifting primitive values into the language, construction and access to pairs, and case expressions. We elide discussion of bindings and application, as it is standard and does not affect the technique. In this section we present a representation for this core language, and show how user-defined algebraic data types can be minimally encoded in it.

The terms of this language can be implemented in Haskell via the data type shown in Fig. 5, which is defined as a generalised algebraic data type (GADT) [26, 38] that adds a *type level* index to the expression syntax tree. Defining  $Exp\ t$ , rather than  $Exp$ , denotes that evaluating the expression yields a value of type  $t$ , which is checked during compilation of the host program. This intrinsically typed representation ensures that embedded language programs written by the user, as well as AST transformations written by the language implementer, are type correct by construction. While choosing a typed representation is a largely orthogonal property of the program representation and makes the techniques we present here more involved, it is well worth the cost for complex embedded languages [7, 20, 21].

The user-facing aspect of the embedded language are terms of type  $Exp\ a$ . These terms are parameterised by the

*surface* type  $a$  of the value that term represents—that is, the user’s view of a data type—just as in the previously discussed examples. The type class [16, 25] `Elt` characterises the *extensible* set of surface types expressible in the language. Its associated type [8, 32] `EltR` maps the surface type to a corresponding internal representation type in terms of the *closed* set of primitive types of the language, `Unit` `()`, `Pair` `(,)`, and `Rec`, with `fromElt` and `toElt` converting between the two.

```
class Elt a where
  type EltR a
  fromElt :: a → EltR a
  toElt   :: EltR a → a
  traceR  :: [TraceR (EltR a)] -- Explained in §5.5
```

It is up to the designer of the embedded language to decide the set of primitive types supported by their language, but this choice does not affect the techniques presented here. An embedded language with built-in support for product types and mutable references, for example, is equally amenable to our technique, which is to enable *user-defined* algebraic data types built from nested pairs of the types in this set.

Instances of the `Elt` class are automatically derivable via GHC Generics [12] (for non-recursive data types). We will show some concrete examples of how algebraic data types are encoded in this representation in the following sections.

#### 4.1 Representation of Product Types

Data types in our internal representation are either constant values introduced by the constructor `Const`, which lifts supported values from the host language into the embedded language, or nested tuples thereof. Nullary and binary tuples are represented by the constructors `Unit` and `Pair` respectively, with `PrjL` and `PrjR` to project the first and second component of a pair.

We represent surface level product types by isomorphic nested pair types. As mentioned previously this mapping from surface to representation type is accomplished via the type class `Elt` and its associated type `EltR`. For example, the surface triple type  $(a, b, c)$  is represented internally by the type  $(((), \text{EltR } a), \text{EltR } b), \text{EltR } c)$ . Primitive types have identity representations. Similarly, user-defined product types such as:

```
data V2 a = V2 a a
data Point = Point Float Float
```

are mapped to the following nested pair types:

```
EltR (V2 a) = (((), EltR a), EltR a)
EltR Point  = (((), Float), Float)
```

In addition to the associated type `EltR` which *deeply* maps a surface type into nested pairs of primitive types, the type class `IsTuple` and associated type `TupleR` perform this mapping to binary pairs at only a *single* level. For example, the surface triple type  $(a, b, c)$  is represented by the nested

pair type  $(((), a), b), c)$ . As with the type class `Elt`, instances of the class `IsTuple` and its associated type `TupleR` are derived automatically. These two classes define how to map an arbitrarily nested data type into its constituent primitive values (`Elt`),<sup>1</sup> and how to project components out at each level of that nesting (`IsTuple`). Continuing the example:

```
TupleR (V2 a) = (((), a), a)
TupleR Point = (((), Float), Float)
```

A value of type `Point` can be lifted from the host language into the embedded language by representing it with the following abstract syntax term:<sup>2</sup>

```
liftPoint :: Point → Exp Point
liftPoint (Point x y) = Tuple $
  Unit `Pair` Exp (Const x) `Pair` Exp (Const y)
```

while extracting the  $x$ -component of an embedded `Point` is represented by:

```
xcoord :: Exp Point → Exp Float
xcoord p = Prj (PrjL (PrjR PrjZ)) p
```

As we can see in `xcoord`, the decomposition of embedded values *adds* embedded language terms. To extract the individual fields of a constructor, we cannot pattern match directly on the `Pair` term.<sup>3</sup> We return to this example in Section 5.1.

#### 4.2 Representation of Sum Types

In contrast to product types, which can be mapped directly to nested pairs, it is not obvious what the internal representation of sum types should be. To implement embedded function `fromMaybe'` from Section 2 we require: (a) a representation for values of type `Exp (Maybe a)`; and (b) a way to distinguish whether the embedded term represents the `Nothing` or `Just` constructor. Meeting the second requirement in such a manner that *embedded* terms can be used in pattern matching in the *host* language is the crux of the challenge.

Recall that during the runtime of the host language program the AST representing the embedded program will be generated, compiled, and executed, before returning the result back into the host program. Thus we can only know which pattern match will succeed *after* the abstract syntax for the program has been generated and the argument evaluated. But we can only generate this abstract syntax by exploring the right-hand-side of the function, which requires matching on the argument and must surely happen *before* executing the program. Pattern matching on embedded sum types therefore would require mixing two distinct

<sup>1</sup>Deeply converting a surface type to nested pairs of primitive types is necessary for external code generation, for example.

<sup>2</sup>We could also write `liftPoint = Const . fromElt` but the chosen exposition allows us to demonstrate how abstract syntax fragments are combined to produce terms of product type.

<sup>3</sup>For example, the function `liftPoint` as defined in the body text or in footnote 2 produce different abstract syntax trees to represent the same value, and we would like that our `xcoord` function work given any type correct term.

phases of program execution: host program runtime (which performs the pattern matching) and embedded program runtime (when the values to be matched are available).

We break this cycle in two parts. First, by having a *uniform* representation for the top-level constructor of each alternative in an embedded sum type; and second, by the method we will use to inspect terms of this type that we will introduce in Section 5.2. For example, the following familiar data types are mapped to this uniform representation as:

```
Eltr Bool          = (TAG, ())
Eltr (Maybe a)    = (TAG, (((), Eltr a))
Eltr (Either a b) = (TAG, ((((), Eltr a), Eltr b))
```

The left component of the pair is a tag—such as `Int`—which indicates which constructor of the sum type the value represents, while the right component of the pair contains the values associated with every constructor of that type.<sup>4</sup>

We can lift a value of type `Maybe Float` from the host into the embedded language with the following abstract syntax:

```
liftMaybe :: Maybe Float → Exp (Maybe Float)
liftMaybe Nothing = Tuple $
  Exp (Const 0) `Pair` (Unit `Pair` Exp undef)
liftMaybe (Just a) = Tuple $
  Exp (Const 1) `Pair` (Unit `Pair` Exp (Const a))
```

where the function `undef` generates the abstract syntax for an undefined value of the required type.<sup>5</sup> Note that this is not the Haskell term `undefined`, rather it is an embedded term that represents a real value with an unspecified bit pattern.<sup>6</sup>

Even with this uniform representation we cannot implement pattern matching by directly inspecting the tag value of a term. Although `liftMaybe` produced tags as constant values, in general these are ASTs representing arbitrarily complex expressions. We address this problem in Section 5.

### 4.3 Representation of Recursive Types

Finally, we outline how to encode terms representing recursive data types. We take the standard iso-recursive approach of treating the recursive type and its one-step unfolding as different, but isomorphic [29]. For example, for the `List` data type we obtain the following representation type, where `Rec` indicates the location of the recursive type:

<sup>4</sup>Alternative representations, such as a tagged pointer to an object containing only the data associated with that particular constructor are of course possible, but that choice is orthogonal to the contribution of this work. We chose a flattened representation of primitive types to (a) simplify the internal language, eliminating the need to deal with references; and (b) because our motivating use case is for data-parallel arrays, for which unboxed data without pointers is the best representation for performance applications and constrained devices such as GPUs (§6).

<sup>5</sup>Values associated with the non-represented alternative contain undefined values, but this is safe as our automatically generated pattern synonyms will never inspect those values (§5.2).

<sup>6</sup>Since the type class `Elt` tells us how to map every supported surface type into nested pairs of primitive types, we can always generate an unspecified value of the appropriate type, for example by choosing zero everywhere.

```
data List a = Nil | Cons a (List a)
Eltr (Rec a) = Rec a
Eltr (List a) = (TAG, ((((), Eltr a), Rec (List a)))
```

In expressions of recursive type, the recursive term is wrapped in the `Roll` instruction (e.g., the `Cons` constructor at the term representing the list tail), and `Unroll` is used to inspect a recursive term. Although this may seem burdensome, in practice these annotations are hidden by our embedded pattern synonyms, which we describe in the following section.

## 5 Implementation

Now that we have introduced our minimal embedded language and demonstrated how to encode algebraic data types in it, we show how we can use pattern synonyms to conveniently work with embedded terms of algebraic data type, and how to integrate this with pattern matching in the host language.

### 5.1 Embedded Products

Recall from Section 4.1 the data type `Point` and its lifting into our expression language. In the host language we pattern match on the constructor `Point` to access the *x*- and *y*-components. To achieve this same functionality in the embedded language, we need a pattern synonym with type:<sup>7</sup>

```
pattern Point_ :: Exp Float → Exp Float
              → Exp Point
```

As both constructing a term of type `Exp Point` as well as destructing these terms to access the stored components amounts to adding new abstract syntax, we require an explicitly bidirectional pattern synonym. The builder function, which specifies how the synonym should behave as an expression, is straightforward:

```
buildPoint :: Exp Float → Exp Float → Exp Point
buildPoint x y = Tuple $
  Unit `Pair` Exp x `Pair` Exp y
```

as is the matcher function, which extracts each component:

```
matchPoint :: Exp Point → (Exp Float, Exp Float)
matchPoint p = ( Prj (PrjL (PrjR PrjZ)) p
               , Prj (PrjR PrjZ) p )
```

Note how the matcher returns its two expression language fragments in a regular Haskell pair. The completed pattern synonym is then:

```
pattern Point_ x y ← (matchPoint → (x, y))
  where Point_ = buildPoint
```

We can now use pattern matching in the embedded language in much the same way as in the host language, for example:

<sup>7</sup>While we make use of the *type* `Point`, we do not use the host language *constructor* `Point`. This is precisely what the pattern synonym provides a replacement for in embedded language code.

```
addPoint :: Exp Point → Exp Point → Exp Point
addPoint (Point_ x1 y1) (Point_ x2 y2) =
  Point_ (x1 + x2) (y1 + y2)
```

Furthermore, due to our uniform representation, defining embedded pattern synonyms for all product types works in the same way. We can abstract this procedure into the following polymorphic pattern synonym:

```
pattern Pattern :: IsPattern s r ⇒ r → Exp s
pattern Pattern r ← (matcher → r)
  where Pattern = builder

class IsPattern s r where
  builder :: r → Exp s
  matcher :: Exp s → r
```

The class `IsPattern` states that we can consider values of type `s` (the user’s data type) as embedded language terms by representing them as we would terms of type `r`. We can write instances for this class that will cover any type `r` whose *representation* type `Eltr r` is equivalent to ( $\sim$ ) the representation type of `s`. For example, the following instance covers any type whose representation is isomorphic to a pair:

```
instance ( Eltr s ~ Eltr (a, b)
         , TupleR s ~ TupleR (a, b) )
  ⇒ IsPattern s (Exp a, Exp b) where
  builder (x, y) = Tuple $
    Unit `Pair` Exp x `Pair` Exp y
  matcher s = ( Prj (PrjL (PrjR PrjZ)) s
             , Prj (PrjR PrjZ) s )
```

Note the similarity of this instance declaration to our specialised functions `buildPoint` and `matchPoint`. The embedded language author defines appropriate instances of this class *once*—for pairs `(,)`, triples `(,,)`, and so forth—and the embedded language user can then define pattern synonyms for their (product) data types without needing to explicitly write the required builder and matcher functions. Our example can now be defined simply as:

```
pattern Point_ x y = Pattern (x, y)
```

## 5.2 Embedded Sums

We demonstrated how pattern synonyms can be used to provide pattern matching for product types in the embedded language. This was possible because pattern synonyms for product types never need to inspect their argument. In this section we discuss the treatment of *sum* data types—types with more than one constructor.

In Section 4.2 we presented the representation for sum data types in the embedded language, and the function `liftMaybe` was an example of how to construct expressions in this form. The challenge is how to treat these values when they are used in patterns. Consider the following embedded function:

```
simple :: Exp (Maybe Int) → Exp Int
simple Nothing_ = 0
simple (Just_ x) = x
```

This gets de-sugared by the host language compiler to a (host language) lambda abstraction of the form:

```
simple = \p → ...
```

The lambda abstraction needs to construct a suitable AST. For that, it cannot inspect its argument `p` as it does not have the actual value that this function will be applied to—that will only be available much later, during embedded program execution. It can only embed `p` as it is into the AST.

To generate the abstract syntax for this function, we need a matcher function which is essentially able to traverse below the lambda abstraction for each equation in the case distinction and extract the corresponding continuation; either the constant value zero, or an expression which extracts the value from the `Just` constructor.

The key idea for the matcher function is that we need to evaluate the function `simple` *twice* to construct the corresponding embedded program. While we don’t have the actual value that this function will be applied to, we can craft a dummy argument that forces a specific pattern in the *host* language to succeed. This way, we will be able to explore each of the right-hand-sides in turn, and construct a program with an *embedded* case statement from those fragments.

For this to work, the construction of the AST for the embedded program has to be a pure function. This is straightforward in our example host language, Haskell, as the type system ensures there are no side effects that might make this technique unsafe, but in less idealistic languages this is the responsibility of the embedded language developer. The restriction of the AST construction to pure functions does not preclude the embedded language itself from containing side effects, however.

**5.2.1 First Steps.** Let us illustrate the problem by first writing the necessary matcher function manually, before we discuss in Section 5.5 this how can be generated automatically. The matcher function has the following type:

```
match_simple1 :: Exp (Maybe Int) → Exp Int
```

This function needs to perform the procedure we outlined, evaluating the function `simple` twice to extract the `Nothing` and `Just` continuations. It has to return a new embedded function of the same type as `simple`, but where the case statement has been lifted from the *host* language into the *embedded* language. This is where we need our two new language terms: `Match` and `Case` (Fig. 5).

Intuitively, `Match` bundles a host language pattern term (its first argument) for each alternative with the actual function (`simple` in our example). The first argument of `Case` is the embedded language term we match on, and a list of expressions constructed via `Match`, paired with exactly the same

trace stored in the `Match`. These constructors help us perform the pattern matching and construct the embedded case expression, respectively. We can implement the matching function for our example as:

```
match_simple1 = \p →
  let rhs_nothing = simple (Match ⟨TRACE 0⟩ p)
      rhs_just    = simple (Match ⟨TRACE 1⟩ p)
  in
  Case p [ (⟨TRACE 0⟩, rhs_nothing)
          , (⟨TRACE 1⟩, rhs_just) ]
```

The constructor `Match` will be consumed by the embedded pattern synonym (it will not be present in the abstract syntax of the embedded program) and is parameterised by a *trace* that will cause a particular host language pattern match to succeed. We give the precise definition of `⟨TRACE⟩` in Section 5.3. This produces the following abstract syntax for the function `simple`:

```
\p → Case p
      [ (⟨TRACE 0⟩, Const 0)
      , (⟨TRACE 1⟩, Prj (PrjR (PrjR PrjZ)) p) ]
```

The interesting part of the `match_simple` implementation is the use of the `Match` constructor, which is the “dummy argument” we use to explore the pattern match alternatives. To see how this works, we turn our attention to the definition of the embedded pattern synonyms for `Maybe`.

**5.2.2 Embedding `Maybe`.** As an example, consider defining the embedded pattern synonym for the `Just` constructor. The builder function is straightforward; it bundles the supplied term together with a concrete tag value indicating which constructor the argument is associated with:

```
buildJust :: Exp a → Exp (Maybe a)
buildJust x = Tuple $
  Exp (Const 1) `Pair` (Unit `Pair` Exp x)
```

The corresponding matcher function is more interesting. Let us first look at its type signature:

```
matchJust :: Exp (Maybe a) → Maybe (Exp a)
```

If the return value is `Nothing` this indicates that the pattern match failed, and if it is `Just` that the pattern match succeeded. Moreover, this type states that, given a term in the *expression* language of type `Maybe a`, it may return in the *host* language an expression of type `Exp a`. Notice how the result of this function is required at host runtime, before the value of the embedded term will be known.

Instead of pattern matching against the value of the argument (which we do not yet have) or against the argument AST itself (which may be arbitrarily complex) we match against a known value: our term wrapped in the `Match` constructor. It is safe to inspect the argument expression only in this case because our `Match` constructor is ephemeral. That is, it is not part of the program the user writes, rather the embedded language author inserts and consumes it during

embedded program construction whenever embedded pattern matching is performed, as sketched in `match_simple1`.<sup>8</sup> Here is a possible (not yet complete) implementation:

```
matchJust1 x = case x of
  Match ⟨TRACE 1⟩ a → Just
                    (Prj (PrjR (PrjR PrjZ)) a)
  Match _         _ → Nothing
```

If the argument is the `Match` term with the constructor trace corresponding to the alternative we are interested in (here `⟨TRACE 1⟩`), then we signal to the host language pattern matcher that the match was a success by returning the abstract syntax that can be used to access the value stored in the embedded `Just` in the success continuation. If the `Match` term does not have the constructor trace we are interested in, we signal to the host language that pattern matching failed by returning `Nothing`. If the argument is not a `Match` term, this corresponds to a usage error, in which case we return an informative error message. In the following section we discuss what this trace structure should be in order to support the required behaviour, and complete the definition of `matchJust`.

### 5.3 Nested Patterns and `⟨TRACE⟩`

Now that we have looked at the key idea behind our approach, let us see how this would work for nested pattern matching. Consider the following example:

```
nested1 :: Exp (Maybe Bool) → Exp Int
nested1 Nothing_      = 0
nested1 (Just_ False_) = 1
nested1 (Just_ True_)  = 2
```

The semantics of pattern matching (§3.2) dictates that for a pattern match to succeed, matching on the pattern  $P$  as well as *all* of the sub-patterns of the pattern must succeed. We do not receive information from the host language pattern matcher as to *why* a pattern succeeded (did it match on the pattern we are interested in, or the wildcard pattern ‘`_?`’) and conversely if it failed, we do not have the opportunity to recover from that failure (for example, by only at that point attempting sub-pattern matches).

To support nested pattern matching, we therefore need a representation of the `⟨TRACE⟩` which specifies a complete pathway of pattern matching through all of the sub-patterns

<sup>8</sup>Since `Match` is used only when constructing the AST of the embedded program, the trace information it contains could be encoded as an annotation over the AST rather than as a term in it. For example, in `Accelerate` the term language is defined using open recursion, so we could include this extra information in the fixed point. This is a technique we use in other parts of the compiler. We did not use that method here to (a) simplify the explanation; and (b) because that makes a *global* change to the data type(s) the compiler deals with, rather than introducing only *local* modifications to how terms are handled in a few key areas. Moreover, this constructor is not needed after we translate from the higher-order to first-order representation in the first phase of our compiler, so this simpler approach is also more appropriate in our particular use case.

supported by this type. We use the following data type as both a witness to the structure of a term, and to denote the alternative chosen at each sub-pattern match:

```
data TraceR a where
  TraceRunit  :: TraceR ()
  TraceRprim  :: PrimType a → TraceR a
  TraceRpair  :: TraceR a → TraceR b
                 → TraceR (a, b)
  TraceRrec   :: TraceR (EltR a) → TraceR (Rec a)
  TraceRtag   :: TAG → TraceR a → TraceR (TAG, a)
```

The first four constructors of this data type act as witnesses to the structure of the representation type of the term. The last constructor indicates the position of a concrete tag value used to discriminate alternatives of a sum data type. For example, the representation type of the surface type `Maybe Bool` is:

```
EltR (Maybe Bool) = (TAG, ((), (TAG, ())))
```

And the value `Just False` has the following trace:

```
TraceRtag 1 `TraceRpair` (TagRunit
  `TraceRpair` (TraceRtag 0
    `TraceRpair` TagRunit))
```

The trace structure allows us to propagate an appropriate sub-pattern trace to every field of the constructor, enabling the host language to explore nested pattern matches. We can now complete the matcher function for the `Just` constructor:

```
matchJust :: Exp (Maybe a) → Maybe (Exp a)
matchJust x = case x of
  Match (TraceRtag 1 (TraceRpair TraceRunit s)) a
    → Just (Match s (Prj (PrjR (PrjR PrjZ)) a))
  Match _ _ → Nothing
```

Adding the `Match` term in the success continuation ensures that nested patterns are in the right form for the embedded pattern synonyms whenever the host language pattern matcher attempts to match on the constructor sub-patterns. The complete embedded pattern synonym is:

```
pattern Just_ :: Exp a → Exp (Maybe a)
pattern Just_ x ← (matchJust → Just x)
  where Just_ = buildJust
```

Finally, notice the pattern match on the result of the view function `matchJust`: this is how the pattern synonym signals to the host pattern matcher the success or failure of the embedded pattern match.

Although this encoding may seem complex, it is necessary for supporting nested pattern matching while also ensuring type safety of the embedded language. Moreover (a) the trace structure can be removed from the final embedded language encoding to use nested case statements directly on literal TAG values (§5.5.1); and (b) these pattern synonyms are generated automatically using `TemplateHaskell` [33]. The embedded language user simply writes the following to have their data type available for use in embedded code:

```
mkPattern ' 'Maybe
```

## 5.4 Embedded Recursive Types

Now that we have introduced embedded pattern synonyms for nested sum and product types, we discuss treatment of recursive types. This works similarly to what we have seen so far, with the addition that we must also insert a `Roll` or `Unroll` term to decorate the recursion point. However, there is one key difference. Consider the following function:

```
nested2 :: Exp (List Int) → Exp Int
nested2 (Cons_ _ (Cons_ _ Nil_)) = 1
nested2 _                        = 0
```

In order to extract the right-hand-side we evaluate the function multiple times, providing the `Match` constructor as argument with a different `<TRACE>` each time. However, the function `nested2` might have any number of specialisations for lists of different lengths, and we receive no feedback as to why a pattern match succeeds or fails, so how do we decide when to stop searching for more unique right-hand-sides?

Although continuing to some arbitrarily chosen depth will catch many common cases—for example, matching on a single-element list—there is in general no solution to this limitation of pattern synonyms in Haskell. To avoid this problem, our embedded pattern synonyms do not provide sub-pattern matches against recursive types—thereby leading to a runtime error if the user attempts to do so—and instead requires them to inspect any recursive terms within a separate case statement.

## 5.5 Embedding Pattern Matching

With our newly defined embedded pattern synonyms, we now only need some way to use them such that the user can—as seamlessly as possible—reuse the host-language pattern matching facilities inside their embedded program. That is, the user should not have to write the matching functions manually, as we did in `match_simple1`. Instead, we provide a higher-order function `match` which does that automatically, such that our example `match_simple1` becomes simply:

```
match_simple2 :: Exp (Maybe Int) → Exp Int
match_simple2 = match simple
```

We can generalise the procedure we followed to write the code for `match_simple1` into the following steps:

1. *Enumerate traces*: For an expression of type `Exp a`, which will be the scrutinee of a case statement or pattern match, we enumerate all possible pattern matches at the type `a`. For example, for the type `Maybe Bool` there are three possibilities: `Nothing`, `Just False`, and `Just True`. These trace alternatives are represented by the `TraceR` structure and generated by the `traceR` function from the `Elt` typeclass.
2. *Collect pattern alternatives*: Given a function `f` of type `Exp a → b`, we apply `f` to each trace alternative generated from the previous step to extract the success continuation

associated with that tag. Apply steps 1 and 2 recursively on  $b$  for  $n$ -ary functions.

3. *Case introduction*: The right hand sides extracted by the previous step are combined into an embedded Case term. At this point we can reintroduce nested case statements and default cases to remove redundant branches.

We will use the `match_simple2` example to explain how the implementation shown in Fig. 6 generates embedded pattern matches in two phases. We have that:

```
match simple = mkFun (mkMatch simple) id
```

First, `mkFun` collects all of the arguments to the function as a typed list of expressions of type `Args`. This argument list is collected by building the continuation `k`, consuming the arguments to the function left-to-right, one binder at a time ②. Once the function is fully saturated this structure is then passed to the function `f` to construct the final embedded term ①. In our example, at this point we have the argument list `x :→ Result`, where `x` was the single argument of type `Exp (Maybe Int)` passed to the function, and `f` is the function `mkMatch simple`.

Second, `mkMatch` lifts a function over expressions (such as `simple`) to a corresponding function that expects its arguments stored in this `Args` list. For each argument in this list it generates all of the pattern alternatives and combines them into an embedded Case term, applying this to the function `f` as it goes to build the final expression. In our example, at ③ we have that `f` is the function `simple` and `x` the single argument to that function.

It is important to first check whether the argument `x` is already in Match form ④, which can arise if `match` is applied multiple times. This check prevents generating an exponential number of Case terms in the embedded code. For every trace supported by this type ⑧—in this example corresponding to `Nothing` and `Just`—we wrap the argument `x` in the proxy Match term ⑨, enabling the corresponding embedded pattern match to succeed. Recursing on the remaining arguments extracts all of the equations of the argument function `f` ⑦. Finally, we introduce the embedded Case term on the scrutinee `x` and list of success continuations `rhs` ⑥. If the type `e` has a single trace alternative—primitive types such as `Int` and data types with a single constructor—this can be elided ⑤. Note that the list `rhs` is in the order defined by the `traceR` function, *not* the order that these equations appear in the user’s source program.

Arguably, one limitation of this approach is the need for `match` to be supplied a *function* on embedded terms, so that it can be continually reapplied to explore every success continuation. For example, the user can not write their embedded program as:

```
inline1 x = case f x of
  Nothing_ → ... -- error: embedded pattern synonym...
  Just_ y   → ... -- ...used outside of 'match' context
```

```
match :: Matching f => f -> f
match f = mkFun (mkMatch f) id

data Args f where
  (:→) :: Exp a -> Args b -> Args (Exp a -> b)
  Result :: Args (Exp a)

class Matching a where
  type Result a
  mkFun    :: (Args f -> Exp (Result a))
            -> (Args a -> Args f) -> a
  mkMatch :: a -> Args a -> Exp (Result a)

instance Elt r => Matching (Exp r) where
  type Result (Exp r) = r
  mkFun f k          = f (k Result) -- ①
  mkMatch e Result = e

instance (Elt e, Matching r)
=> Matching (Exp e -> r) where
  type Result (Exp e -> r) = Result r
  mkFun f k = \x ->
    mkFun f (\xs -> k (x :→ xs)) -- ②
  mkMatch f (x :→ xs) = -- ③
    case x of
      Match _ _ -> mkMatch (f x) xs -- ④
      _         -> case rhs of
        [(_,r)] -> r -- ⑤
        _       -> Case x rhs -- ⑥
    where
      rhs = [ (trace, mkMatch (f x') xs) -- ⑦
             | trace ← traceR @e -- ⑧
             , let x' = Match trace x ] -- ⑨
```

**Figure 6.** Implementation of the `match` function, which enables us to lift pattern matching from the host language to the embedded language.

Since the `match` function is not invoked in `inline1` neither of the pattern matches will succeed. However as embedded pattern synonyms require their argument to be in the form generated by `match`, we can detect this usage mistake and present the user with an informative error message. In order to still make use of inline case statements we rewrite this example as:<sup>9</sup>

```
inline2 x = f x & match \case
  Nothing_ → ...
  Just_ y   → ...
```

<sup>9</sup>Using the syntactic extensions `lambda case` and `block arguments`, together with the reverse application operator `(&) :: a -> (a -> b) -> b` from the standard module `Data.Function`.

**5.5.1 Nesting case Statements.** As discussed, due to the semantics of pattern matching the `TraceR` structure necessarily represents an entire pathway through all of the sub-patterns of a constructor that result in a successful pattern match. Therefore, scrutinising a term of type `Maybe Bool`, for example, yields a `Case` term with three success continuations. Similarly, we can not know if a pattern match succeeded because it matched on the wildcard pattern `'_'`, which can lead to redundant equations in the list of success continuations.

These limitations are unavoidable during host program generation of the embedded program. Only once the abstract syntax for the embedded program has been generated using the above procedure can we optimise it by (a) introducing nested case statements, for example by the method of Wadler [39] or Augustsson [2], and; (b) introducing default statements to eliminate redundancy, for example by directly comparing terms for equality. This is fairly straightforward and, moreover, orthogonal to the contribution presented in this work, so we elide the details; the interested reader may refer to our implementation of this technique in the embedded language `Accelerate` for a complete example (§6).

## 5.6 Embedded Products, Revisited

As a final detail, we return to our formulation of embedded pattern synonyms for product types (§5.1). Consider:

```
nested3 :: Exp (Maybe Int, Bool) → Exp Int
nested3 x = x & match \case ...
```

When used within a `match` context (the argument is wrapped in a `Match` term), the pair constructor must propagate the sub-pattern traces to each component of the pair. Outside of this context the `Match` term can be ignored. Similarly, the `traceR` instance for product types returns the Cartesian product of the sub-pattern traces.

## 6 Case study

We implemented embedded pattern matching in `Accelerate` [7, 10, 20, 21, 37], an open source language deeply embedded in Haskell for data-parallel array computations. `Accelerate` includes a runtime compiler which generates and compiles parallel code using LLVM [17]. Adding embedded pattern matching to `Accelerate` allowed us to simplify both the internals of the compiler, for example by removing `Bool` as a primitive type, as well as the user facing language. The `stimes` function from the `Semigroup` instance changed from:

```
stimes n x = lift . Sum $
  fromIntegral n * getSum (unlift x :: Sum (Exp a))
```

to code that, apart from a trailing underscore, is exactly the same as the definition of this procedure in the host language:

```
stimes n (Sum_ x) = Sum_ $ fromIntegral n * x
```

While we did not run a controlled user study to systematically evaluate the effect of these changes on the usability of the language, we do have anecdotal evidence that it has

improved the experience for users. We have been using `Accelerate` in a bachelor course for parallel and concurrent programming with about 200 students. In the first iteration of the course we used the original version of `Accelerate`, without embedded pattern matching, and in the following years we used `Accelerate` with our changes implemented. The cohort who used the original version struggled to understand when and how to use `lift` and `unlift` to work with algebraic data types, and found it difficult to interpret the error messages that the compiler emitted when these operations were missing or used incorrectly. These error messages are difficult to understand (as they expose details of the implementation) and difficult to fix (as they require knowledge of extensions such as `ScopedTypeVariables`). The later cohorts experienced none of these issues: once they were told how to use pattern synonyms—for example to use `T2` to construct and deconstruct pairs—they required no further assistance.

For algebraic data types with a single constructor the code that is generated by the compiler is the same as it was before our changes. However, for sum data types we are now able to emit `switch` instructions directly. This is an example of a situation where the built-in notion of a case statement makes it possible to generate more efficient code, for example by producing a lookup table (a single load instruction) instead of a sequence of conditional branch instructions. This also demonstrates why our representation for sum types uses a flattened structure with a single `TAG`, rather than a nested binary `Either` type analogously to how product types are represented as nested binary pairs.

## 7 Related Work

In this paper we used the example of an embedded language constructed in an *initial* encoding by representing its abstract syntax as data constructors (§4). An alternative is to use a *final* encoding of the abstract syntax as combinator functions [5], which can allow more flexibility in the interpretation of a language built in this style, but does not change what features of the host language can be captured in that representation.<sup>10</sup> It should therefore be possible to combine that embedding approach with our technique.

There are several ways to *lift* values from the host language into the embedded language. The approach we demonstrated here also requires some method to *unlift* values, which is possible for product data types, but not for sum data types. For example, to access each component of an expression of type `Exp (a, b)`, it would first be unlifted into a pair of expressions of type `(Exp a, Exp b)`. Svenningsson [34] always use this latter form, which avoids the unlifting step but has the disadvantage of duplicating embedded terms and requires common-subexpression elimination during code generation to recover shared structures. Both of

<sup>10</sup>At least for staged/code generating interpretations of the encoding, which imposes the most restrictions and is the target of this work.

these approaches have the subtle limitation that the host language container data type must be polymorphic in *all* of its arguments. Our approach of using pattern synonyms does not have this limitation, and supports non-polymorphic data types such as the `Point` type we saw earlier.

Our technique is implemented purely within the concrete syntax of the host language,<sup>11</sup> but if we are willing to surrender this quality then other approaches become possible. QDSLs [24] make use of the quasi-quoting [18] feature of TemplateHaskell [33] to allow the embedded language author to define their own syntax (and associated parser) for their embedded language, rather than reusing the syntax of the host language. Quoting is indicated by `[|...|]` and unquoting by `$(...)`, which adds syntactic noise but clearly demarcates between host and embedded code. That work demonstrates a QDSL-based version of Svenningsson's [34] embedded language, but unfortunately does not explore how quotation might overcome that language's limitations. The LINQ framework as used in C# [23] and F# [35], and the Lightweight Modular Staging (LMS) [31] framework in Scala, also make use of quotation techniques.

Haskino [15] goes further and uses a GHC plugin to directly manipulate the AST of the host program during Haskell compilation time. As with QDSLs, changing or directly manipulating the host language provides more opportunities for adding embedded language features, but as we have shown in this work, such drastic measures are not always required.

## 8 Conclusion

We have presented a method for pattern matching on user-defined algebraic data types in embedded languages. This improves the usability of embedded languages—at last bringing a distinctive and arguably defining feature of functional programming to embedded languages—and, potentially, improving the quality of code generated by the embedding. While the syntactic integration between the host and embedded languages depends on features of the host language—we make extensive use of pattern synonyms provided by languages such as Haskell and Scala, for example—our key contributions showing how to generate embedded case expressions can be implemented in any general purpose host language. Although the AST generation has to be side-effect free, as it may be executed several times, neither the host nor embedded languages themselves must be side-effect free.

## References

- [1] Robert Atkey, Sam Lindley, and Jeremy Yallop. 2009. Unembedding domain-specific languages. In *Haskell Symposium*. <https://doi.org/10.1145/1596638.1596644>
- [2] Lennart Augustsson. 1985. Compiling pattern matching. In *FPCA: Functional Programming Languages and Computer Architecture*. [https://doi.org/10.1007/3-540-15975-4\\_48](https://doi.org/10.1007/3-540-15975-4_48)
- [3] Emil Axelsson. 2012. A generic abstract syntax model for embedded languages. In *ICFP: International Conference on Functional Programming*. <https://doi.org/10.1145/2364527.2364573>
- [4] Emil Axelsson, Koen Claessen, Mary Sheeran, Josef Svenningsson, David Engdal, and Anders Persson. 2011. The Design and Implementation of Feldspar: An Embedded Language for Digital Signal Processing. In *IFL: Implementation and Application of Functional Languages*. 121–136. [https://doi.org/10.1007/978-3-642-24276-2\\_8](https://doi.org/10.1007/978-3-642-24276-2_8)
- [5] Jacques Carette, Oleg Kiselyov, and Chung-chieh Shan. 2009. Finally Tagless, Partially Evaluated: Tagless Staged Interpreters for Simpler Typed Languages. *Journal of Functional Programming* 19, 5 (2009), 509–543. <https://doi.org/10.1017/S0956796809007205>
- [6] Manuel Chakravarty, Roman Kireev, Kenneth MacKenzie, Vanessa McHale, Jann Müller, Alexander Nemish, Chad Nester, Michael Peyton Jones, Simon Thompson, Rebecca Valentine, and Philip Wadler. 2019. Functional Blockchain Contract. <https://iohk.io/en/research/library/papers/functional-blockchain-contracts/>.
- [7] Manuel M T Chakravarty, Gabriele Keller, Sean Lee, Trevor L McDonell, and Vinod Grover. 2011. Accelerating Haskell array codes with multi-core GPUs. In *DAMP: Declarative Aspects of Multicore Programming*. <https://doi.org/10.1145/1926354.1926358>
- [8] Manuel M T Chakravarty, Gabriele Keller, and Simon Peyton Jones. 2005. Associated type synonyms. In *ICFP: International Conference on Functional Programming*. <https://doi.org/10.1145/1086365.1086397>
- [9] Koen Claessen and David Sands. 1999. Observable sharing for functional circuit description. In *Asian Computing Science Conference*. [https://doi.org/10.1007/3-540-46674-6\\_7](https://doi.org/10.1007/3-540-46674-6_7)
- [10] Robert Clifton-Everest, Trevor L. McDonell, Manuel M T Chakravarty, and Gabriele Keller. 2017. Streaming Irregular Arrays. In *Haskell Symposium*. <https://doi.org/10.1145/3122955.3122971>
- [11] Conal Elliott. 1997. Modelling Interactive 3D and Multimedia Animation with an Embedded Language. In *Domain Specific Languages*.
- [12] Jeremy Gibbons. 2006. Datatype-generic programming. In *Spring School on Datatype-Generic Programming*. [https://doi.org/10.1007/978-3-540-76786-2\\_1](https://doi.org/10.1007/978-3-540-76786-2_1)
- [13] Andy Gill. 2009. Type-safe observable sharing in Haskell. In *Haskell Symposium*. <https://doi.org/10.1145/1596638.1596653>
- [14] Andy Gill, Tristan Bull, Garrin Kimmell, Erik Perrins, Ed Komp, and Brett Werling. 2009. Introducing Kansas Lava. In *IFL: Implementation and Application of Functional Languages*. [https://doi.org/10.1007/978-3-642-16478-1\\_2](https://doi.org/10.1007/978-3-642-16478-1_2)
- [15] Mark Grebe and Andy Gill. 2016. Haskino: A Remote Monad for Programming the Arduino. In *PADL: Practical Aspects of Declarative Programming*. [https://doi.org/10.1007/978-3-319-28228-2\\_10](https://doi.org/10.1007/978-3-319-28228-2_10)
- [16] Cordelia V Hall, Kevin Hammond, Simon L Peyton Jones, and Philip L Wadler. 1996. Type classes in Haskell. *Transactions on Programming Languages and Systems* 18, 2 (1996), 109–138. <https://doi.org/10.1145/227699.227700>
- [17] Chris Lattner and Vikram Adve. 2004. LLVM: A Compilation Framework for Lifelong Program Analysis & Transformation. In *CGO: Code Generation and Optimization*. <https://doi.org/10.1109/CGO.2004.1281665>
- [18] Geoffrey Mainland. 2007. Why it's nice to be quoted: Quasiquoting for Haskell. In *Haskell Symposium*. <https://doi.org/10.1145/1291201.1291211>
- [19] Simon Marlow. 2010. *Haskell 2010 Language Report*. <https://www.haskell.org/onlinereport/haskell2010/>
- [20] Trevor L. McDonell, Manuel M T Chakravarty, Vinod Grover, and Ryan R Newton. 2015. Type-safe Runtime Code Generation: Accelerate to LLVM. In *Haskell Symposium*. <https://doi.org/10.1145/2804302.2804313>

<sup>11</sup>While pattern synonyms make the integration seamless they are not required for the core of the technique. Host languages without pattern synonyms could integrate this technique via, for example, a macro system.

- [21] Trevor L. McDonell, Manuel M T Chakravarty, Gabriele Keller, and Ben Lippmeier. 2013. Optimising Purely Functional GPU Programs. In *ICFP: International Conference on Functional Programming*. <https://doi.org/10.1145/2500365.2500595>
- [22] Trevor L. McDonell, Joshua D. Meredith, and Gabriele Keller. 2022. Embedded Pattern Matching (Artefact). In *Haskell Symposium*. <https://doi.org/10.1145/3554299>
- [23] Erik Meijer, Brian Beckman, and Gavin Bierman. 2006. LINQ: Reconciling object, relations and XML in the .NET framework. In *SIGMOD: Management of Data*. <https://doi.org/10.1145/1142473.1142552>
- [24] Shayan Najd, Sam Lindley, Josef Svenningsson, and Philip Wadler. 2016. Everything old is new again: quoted domain-specific languages. In *PEPM: Partial Evaluation and Program Manipulation*. <https://doi.org/10.1145/2847538.2847541>
- [25] John Peterson and Mark Jones. 1993. Implementing type classes. In *PLDI: Programming Language Design and Implementation*. <https://doi.org/10.1145/155090.155112>
- [26] Simon Peyton Jones, Geoffrey Washburn, and Stephanie Weirich. 2004. *Wobbly types: type inference for generalised algebraic data types*. Technical Report MS-CIS-05-26.
- [27] Frank Pfenning and Conal Elliott. 1988. Higher-Order Abstract Syntax. In *PLDI: Programming Language Design and Implementation*. <https://doi.org/10.1145/53990.54010>
- [28] Matthew Pickering, Gergő Érdi, Simon Peyton Jones, and Richard A. Eisenberg. 2016. Pattern Synonyms. In *Haskell Symposium*. <https://doi.org/10.1145/2976002.2976013>
- [29] Benjamin C. Pierce. 2002. *Types and Programming Languages*. MIT Press.
- [30] John C. Reynolds. 1972. Definitional Interpreters for Higher-Order Programming Languages. In *ACM Annual Conference*. <https://doi.org/10.1145/800194.805852>
- [31] Tiark Rompf and Martin Odersky. 2010. Lightweight Modular Staging: a pragmatic approach to runtime code generation and compiled DSLs. In *GPCE: Generative Programming and Component Engineering*. <https://doi.org/10.1145/1868294.1868314>
- [32] Tom Schrijvers, Simon Peyton Jones, Manuel M T Chakravarty, and Martin Sulzmann. 2008. Type checking with open type functions. In *ICFP: International Conference on Functional Programming*. <https://doi.org/10.1145/1411204.1411215>
- [33] Tim Sheard and Simon Peyton Jones. 2002. Template meta-programming for Haskell. In *Haskell Workshop*. <https://doi.org/10.1145/581690.581691>
- [34] Josef Svenningsson and Emil Axelsson. 2012. Combining Deep and Shallow Embedding for EDSL. *Trends in Functional Programming* 7829 (2012). [https://doi.org/10.1007/978-3-642-40447-4\\_2](https://doi.org/10.1007/978-3-642-40447-4_2)
- [35] Don Syme. 2006. Leveraging .NET meta-programming components from F#: integrated queries and interoperable heterogeneous execution. In *ML*. <https://doi.org/10.1145/1159876.1159884>
- [36] Emina Torlak and Rastislav Bodik. 2013. Growing Solver-Aided Languages with Rosette. In *Onward! New Ideas, New Paradigms, and Reflections on Programming & Software*. 135–152. <https://doi.org/10.1145/2509578.2509586>
- [37] Lars B. van den Haak, Trevor L. McDonell, Gabriele Keller, and Ivo Gabe de Wolff. 2020. Accelerating Nested Data Parallelism: Preserving Regularity. In *Euro-Par: Parallel Processing*. [https://doi.org/10.1007/978-3-030-57675-2\\_27](https://doi.org/10.1007/978-3-030-57675-2_27)
- [38] Dimitrios Vytiniotis, Stephanie Weirich, and Simon Peyton Jones. 2006. Simple unification-based type inference for GADTs. In *ICFP: International Conference on Functional Programming*. <https://doi.org/10.1145/1159803.1159811>
- [39] Philip Wadler. 1987. Efficient compilation of pattern-matching. In *The Implementation of Functional Programming Languages*, Simon Peyton Jones (Ed.). Prentice Hall, 78–103.
- [40] David Young, Mark Grebe, and Andy Gill. 2021. On Adding Pattern Matching to Haskell-Based Deeply Embedded Domain Specific Languages. In *PADL: Practical Aspects of Declarative Programming*. [https://doi.org/10.1007/978-3-030-67438-0\\_2](https://doi.org/10.1007/978-3-030-67438-0_2)